Tutorial de Prolog
"**Pro***gramming in* **Log***ic*"
(Programando em Lógica)

Ilaim Costa Júnior (ilaim@joinville.udesc.br)
Claudio Cesar de Sá (claudio@joinville.udesc.br)
UDESC/CCT/DCC - Campus Universitário
Bloco F - 3o. Andar - Sala F. Gauss
89.223-100 - Joinville - SC

Versão 2.5
Joinville, 6 de Março de 2003

6 de Março de 2003

Conteúdo

1 Introdução **4**

1.1 Comece aqui 4

1.2 Resolvendo um Problema 5

1.3 Apresentação Inicial 7

1.4 Paradigmas de Linguagens 8

1.5 Máquina "*Prologuiana*" 8

1.6 Exemplos 9

1.7 Sintaxe do Prolog 10

1.8 Detalhando a Sintaxe 11

1.9 Outros Conceitos 12

1.10 Operações Básicas do SWI-Prolog 13

1.11 Outros Detalhes 16

2 Recursividade **18**

2.1 Exemplos sobre Recursividade 20

3 Functores **24**

3.1 Definições 24

3.2 Motivação 25

3.3 Exemplo 25

3.4 Discussão sobre Problemas de IA 26

3.5 Usando Functores em Problemas de IA 27

3.6 Concluindo Functores 27

4 Listas **28**

4.1 Definições 28

4.2 Exemplos 29

4.3 Combinando os Functores as Listas 33

5 Gerando Programas Executáveis (ou quase com o SWI-Prolog) **35**

6 Operações especiais **37**

7 Programando com "*Elegância*" **38**

8 Gerando Sequências de Dominó **38**

9 Sites interessantes **41**

Lista de Figuras

1 Arquitetura Típica de uma Máquina "*Prologuiana*" 9

2 Fluxo "*Aproximado*"de Inferências em Prolog 13

3 Árvore de busca construída pelo Prolog para regra **par** 18

4 Uma árvore que representa os descendentes de alguém 23

5 Um grafo que define custos entres as cidades 23

6 Representando o conceito de hierarquia com um functor 24

1 Introdução

1.1 Comece aqui ...

Antes de iniciar a leitura deste tutorial, assegure que "*tenhas em mãos*" uma das últimas versões de algum Prolog padrão. Caso esta resposta seja negativa, veja a seção 9, onde se encontra uma boa documentação disponível sobre a linguagem Prolog. Há vários Prolog gratuitos na WEB, particularmente, usaremos o SWI-Prolog http://www.swi-prolog.org/, para Windows (95, 98 ou NT) ou Linux ("*Unix-like*"). Recentemente, ficou disponível o **XPCE** (http://www.swi-prolog.org/) que é o SWI-Prolog com recursos gráficos, análogo a qualquer ambiente de desenvolvimento. Com um detalhe: também é gratuito!

Para o curso de Lógica e Programação em Lógica (UDESC), este material pode ser encontrado no seguinte servidor:

\\200.19.107.36\disciplinas\lpl-logica

(inclusive uma versão on-line deste material) a partir de um navegador qualquer em um dos laboratórios do 3o. andar.

Nestas notas, apenas os principais tópicos da linguagem são apresentados, outros detalhes "*o aluno terá que correr atrás*". Ou seja, nestas páginas o aluno tem "**um guia de sobrevivência**" com o Prolog, pois consta a parte mais complicada da linguagem. Os livros e manuais sobre o assunto são indispensáveis, ver a seção 9.

Os termos enfatizados ou em itálicos, são explicados em sala de aula, bem como, parte do material complementar da disciplina. Eventualmente, o aluno deve repassar todas estas notas, antes de insistir numa compreensão sequencial do texto. A propósito, o Prolog não apresenta o conceito de sequencialidade e fluxo linear de programas, a exemplo de outras linguagens tradicionais de programação. Vale a proposta de experimentar, sem se preocupar numa sequencialidade.

Este tutorial encontra-se "*relaxado*" no que concerne a terminologia da lógica (**não é a lógica de programação**, comumente confundida) e do Prolog. Para aqueles que quiserem investigar o Prolog e seus fundamentos lógicos, aconselho o livro: "*Logic, Programming and Prolog*", de Ulf Nilsson e Jan Maluszyns, editado pela by John Wiley & Sons, 2000; mas também disponível gratuitamente em: *http://www.ida.liu.se/ ulfni/lpp/*.

Finalmente, a proposta é tornar o texto acessível a todas as pessoas que não tenham base em lógica, e até mesmo aqueles que nunca tenham utilizado nenhuma outra linguagem de programação.

1.2 Resolvendo um Problema

Esse exemplo é instigante e contém "*todos*" conceitos iniciais do Prolog, acompanhe com atenção a discussão em sala de aula. A proposta é resolver de imediato um problema interessante, antes de entrar nos detalhes da linguagem. Seja o enunciado:

" *Alexandra, Barbara, e Rosa tem roupas em azul, verde, lilás, vermelho e amarelo. Nenhuma delas veste amarelo com vermelho. Cada uma delas veste roupas com duas cores, isto é, roupas tipo blusa e saia. Alexandra está vestindo o azul. Barbara está vestindo o amarelo mas não o verde. Rosa veste o verde mas não se veste nem de azul e nem lilás. Uma delas está de vermelho. Uma das cores é usada tanto por Barbara como por Rosa. Alexandra e Barbara tem 04 cores entre elas.*"

Acompanhe o texto e verifique como o mesmo foi escrito na sintaxe Prolog (a ser apresentada em breve), no programa abaixo:

```
/* Aqui começa o programa ... */

pessoa(ale).    /* Alexandra */
pessoa(babe).   /* Barbara */
pessoa(rosa).

cor(azul).
cor(amarelo).
cor(vermelho).
cor(lilaz).
cor(verde).

/* este pedaço é importante... pois.... está
dito no problema e define uma cor para cada
moça */

tem_uma_cor(babe,amarelo).
tem_uma_cor(rosa,verde).
tem_uma_cor(ale,azul).

/* aqui pode-se ampliar as restrições */
nao_pode_vestir(babe,verde).
nao_pode_vestir(rosa,azul).
nao_pode_vestir(rosa,lilaz).

veste(X, Cor1, Cor2) :-
    pessoa(X),
    cor(Cor1),
    cor(Cor2),
    tem_uma_cor(X,Cor1),
```

```
    \+(nao_pode_vestir(X,Cor2)) ,    /* é um fato que deve ser negado */
    Cor1 \== Cor2 ,         /* C1 diferente de C2 */
    /* não pode ter amarelo e vermelho com C1 e C2 e vice-versa */
    \+(
      ((Cor1 = vermelho) ,  /* C1 tem que ser diferente de Vermelho */
      (Cor2 = amarelo))     /* idem para amarelo.... */
    ;                       /*  ; ou */
      ((Cor1 = amarelo)  ,  /* , e  ... o contrário */
      (Cor2 = vermelho))
    ).

ache_tudo :-
    veste(X,C1,C2),
    write(X),
    write('   ====> '), write(C1),
    write('.... e ....'), write(C2),
    nl,
    fail.

ache_tudo.
```

Feita edição do programa num editor ASCII qualquer, carregue-o num interpretador ou compile-o. Usando o SWI-Prolog, temos:

```
pl ^  /* ativa o interpretador */
.................................................
?- consult('roupas.pl').
% roupas.pl compiled 0.00 sec, 0 bytes
Yes
?-
?- ache_tudo.
ale   ====> azul.... e ....vermelho
....
babe   ====> amarelo.... e ....azul
....
rosa   ====> verde.... e ....amarelo
....
Yes
?-
```

Ao longo deste texto, este exemplo será detalhado, bem como o estudante vai estar apto em resolvê-lo de outras maneiras.

1^o. **Exercício**: Execute este programa, ativando o *trace* gráfico, **guitracer**; e analise porque alguns resultados foram duplicados. Ou seja:

```
?- guitracer.                    /* atriva o trace gráfico */
....
?- trace, ache_tudo.            /* faz uma pergunta com o trace */
```

2^o. **Exercício**: Faça as seguintes experimentações na console do interpretador:

1. ?- pessoa(X).
2. ?- cor(Y).
3. ?- tem_uma_cor(X,Y).
4. ?- nao_pode_vestir(X,Y).
5. ?- veste(X,Y,Z).

Não avance ao exercício seguinte, caso não tenhas entendido o que levou a Máquina Prolog, deduzir tais respostas. Caso tenhas dúvida, habilite o trace (?- trace, ...) no prefixo das questões acima.

3^o. **Exercício**: Altere e/ou inclua algumas regras, afim de restringir a quantidade de respostas.

1.3 Apresentação Inicial

Definição: Prolog é uma linguagem de programação implementada sobre um paradigma lógico. A base de tal lógica é o Cálculo dos Predicados ou Lógica de 1^a. Ordem (LPO). Logo, a concepção da linguagem apresenta uma terminologia própria, mas fundamentada sobre a LPO. As restrições da completude e corretude da LPO são contornadas com um método de prova sistemático e completo. Detalhes: "*Logic, Programming and Prolog*", de Ulf Nilsson e Jan Maluszyns, editado pela by John Wiley & Sons, 2000; e gratuitamente em: *http://www.ida.liu.se/ ulfni/lpp/*.

Características:

- Manipula símbolos (objetos) por natureza, logo "*7 + 13*" pode apresentar várias conotações;
- Átomo, literal ou um objeto: 'a', 77, "aa", '7' , "777"...
- Seu princípio inferencial é o "*backward chaining*" (*encadeamento regressivo*), ou seja, para encontrar algum símbolo como verdade, devemos demonstrar que suas premissas eram também verdadeiras;
- Apresenta um propósito geral como Linguagem de Programação, bem como tem interface com linguagens como Delphi, C, Visual Basic, e outras;
- Portabilidade entre plataformas para o Prolog padrão. Logo, não é uma linguagem proprietária;
- Fácil aprendizado;
- Áreas de aplicações: problemas de IA, provadores de teoremas, sistemas especialistas, pesquisa operacional, construção de compiladores, etc.

1.4 Paradigmas de Linguagens

Há outros paradigmas de linguagens de programação, além do **lógico** cujo principal representante é o Prolog. São eles:

Imperativo : Linguagem C, Pascal, Fortran, ...etc.
Orientação a Objetos : Smalltalk, Simula
Funcional : Lisp, Miranda, Haskell, ML, ...etc.
Híbridas : Delphi, C++, C-Builder, Kylix, ...etc.
Baseada em Regras de Produção : CLIPS, OPSS5, Shell para SE's, ...etc.
Específicas : GPSS, e outras com hardware específicos.

1.5 Máquina "*Prologuiana*"

Uma arquitetura típica dessa linguagem de programação, encontra-se na figura 1.5. Sua fundamentação teórica encontra-se no processo de raciocínio típico do "*backward chaining*". A operacionalidade é a mesma desse esquema de raciocínio, logo, consulte o material em questão, ou não perca as explicações do professor.



Figura 1: Arquitetura Típica de uma Máquina "*Prologuiana*"

Enfim, acompanhe os exemplos que se seguem tendo por base essa figura.

1.6 Exemplos

Além do Prolog instalado, voce precisa tem um editor de texto padrão ASCII, aconselho o **edit** do MS-DOS, ou ainda, o **bloco de notas** do Windows. Caso uses um Prolog com interface gráfica tipo XPCE (http://www.swi-prolog.org/), o qual funciona muito bem sob o Linux, tal editor é acionado pelo comando:

?- edit(file('nome do pgm.pl')).

O ambiente do SWI-Prolog é inicialmente interpretado, e apresenta um "*prompt default*" ao ser executado: **?-**

Veja os exemplos iniciais:

```
?- 2 > 3.
No
?- 2 == 2.
Yes
?- mae(eva, abel).
No
?- 3 > 2.
Yes
```

Caso queiras carregar um programa usuário, previamente editado e salvo no padrão ASCII, para ser usado no SWI-Prolog, então:

```
?- consult('c:/temp/teste.pl').

% teste.pl compiled 0.00 sec, 560 bytes
Yes
?- homem( X ).
X = joao
Yes
?- homem( x ).
No
```

Adiantando, as letras maiúsculas são as **variáveis** da linguagem Prolog, isto é, aceitam qualquer objeto.

1.7 Sintaxe do Prolog

A sintaxe "**Prologuiana**" é simples. Esta é construída a partir de formulações com os predicados lógicos, logo, o conceito de "*predicado*" reflete o "*espírito*" do Prolog.

As construções (*linhas de código em Prolog*) seguem uma sintaxe de três tipos.

1o. tipo: são as **questões** ou *"goals"*, isto é, uma pergunta à uma base de conhecimento (que fica instanciado e referenciado neste ambiente inferencial do Prolog)
. . . "a idéia a de um teorema a ser demonstrado" . . .

```
?-  >(3,2).    /* ou  ?-  3 > 2.      Ou seja,   3  é maior que 2  */
?-
Yes
```

2o. tipo: são os **fatos** ... algo sempre verdadeiro ou *verdades incondicionais*, encontrados na "*base de conhecimento*".

```
?- listing(homem).
homem(joao).
homem(jose).
homem(jedro).
Yes
```

3o. tipo: são as **regras** ... que "*aproximadamente*" são verdades ou teoremas condicionais, isto é: *necessitam de uma prova lógica*!

```
?- listing(mortal).
mortal(A)  :-   homem(A).
/*  obs.     :-   eh o implica ao contrário  <-   */
/*  homem --->  mortal
 leia-se ... para demonstrar que algum X é
 mortal, preciso demonstrar e provar que A eh
 um homen  */
Yes
```

Conclusão: *a construção de um programa em Prolog, é feita de: fatos, questões e regras... e nada mais!*

1.8 Detalhando a Sintaxe

Há alguns detalhes de como se montam esses predicados "*prologuianos*". Essa "*Receita de Bolo*" segue abaixo:

- Os predicados devem ser em letras minúsculas.... *OBRIGATORIAMENTE*!
- Os fatos, regras e questões terminam por "**.**" (ponto);
- Seja uma regra, exemplo: `irmão(X,Y) :- pai(X,Z), pai(Y,Z), X \== Y.` . Os seguintes elementos são identificados:
  - "*irmão/2*", "*pai/2*" e "*\ == /2*", são predicados binários (i. é. aridade igual a dois);
  - A vírgula (",") é o "*and*" lógico;
  - O "*or*" lógico é o ponto-e-vírgula ("**;**") ;
  - O "**:-** " e o "*pescoço*" da regra;
  - O ponto no final "**.**" é o "*pé*" da regra;
  - Logo a regra possui uma "*cabeça*", que é o predicado "*irmão/2*";
  - O corpo é composto pelos predicados: "*pai/2*" e "*\ == /2*".

- Os nomes que começam por letras maiúsculas são variáveis no Prolog. As variáveis podem receber números, letras, frases, arquivos, regras, fatos, ... até mesmo outras variáveis, mesmo sendo desconhecidas !
- A variável anônima "_" (o *"underscore"*) não necessita ser instanciada, i.é. não impõe restrições quanto ao seu escopo;
- Nomes que começam por letras minúsculas são símbolos no Prolog;
- O escopo é a validade da instância de uma variável na regra;
- As variáveis possuem o seu escopo apenas dentro de suas regras;

1.9 Outros Conceitos

1. Aridade: Número de argumentos do predicado. Leia-se que "*capital(paris, França)*" é um predicado de aridade dois;

2. "*Matching*": Ao processar uma busca na *Memória de Trabalho* (MT), a máquina inferência do Prolog realiza verificações do início do programa para o final. Encontrando um predicado de mesmo nome, em seguida verifica a sua aridade. Se nome e aridade "*casarem*", o "*matching*" foi bem sucedido;

3. Instância: Quando uma variável de uma regra for substituída por um objeto, esse é dito ser uma instância temporária à regra;

4. Unificação: Quando uma regra for satisfeita por um conjunto de objetos, se encontra um instância bem sucedida para regra inteira. Neste caso, as variáveis foram unificados pelos objetos, resultando em uma regra verdade e demonstrável. Em resumo, é uma instância que foi bem sucedida;

5. "*Backtracking*": Esse conceito é particular ao Prolog, e o diferencia das demais linguagens convencionais de computador. Basicamente, *e não completamente*, o conceito de "*backtracking*" é o mesmo da Engenharia de Software. Ou seja, na falha de uma solução proposta, o programa deve retroceder e recuperar pontos e/ou estados anteriores já visitados, visando novas explorações a partir destes. Como exemplo ilustrativo, imaginemos uma estrutura em árvore qualquer, e em que um dos nós terminais (folhas) exista uma ou mais saídas. Qual a heurística de encontrar uma das saídas ou o nó desejado? Várias abordagens podem ser feitas, assunto esse que foge da discussão no momento. O Prolog usa uma busca em profundidade ("*depth-first*"), cujo retrocesso ocorre ao nó mais recente, cuja visita foi bem sucedida. Esses nós referem aos predicados que formam um corpo da árvore corrente de busca. Logo, várias árvores são construídas, cada vez que uma nova regra é disparada. Mas o controle dessas árvores e seu objetivo corrente, é implementado como uma estrutura de pilha, em que nós não solucionados são empilhados.

Os conceitos acima, fazem parte de uma **tentativa** de descrever a "**Máquina Prologuiana**", via um fluxograma, veja figura 1.9:

Figura 2: Fluxo "*Aproximado*"de Inferências em Prolog

Verifique se a interpretação desse esquema da *pilha abstrata* de questões que Prolog faz, representado na figura 1.9, foi realmente entendido. Em caso negativo, solicite ao professor a explicação, ou referencie na bibliografia sugerida.

1.10 Operações Básicas do SWI-Prolog

Exceto o Visual Prolog (http://www.visual-prolog.com/), todos os demais Prolog são basicamente idênticos. Atualmente deve existir ter uns dez (10) fabricantes de Prolog comerciais e "*um outro tanto*" oriundos da academia. Todos oferecem algo gratuito como chamariz ao seu sistema.

O SWI-Prolog não foge a regra, as suas características são: velocidade, portabilidade (interoperacionabilidade), padrão, robustez, facilidades de uso, interface com várias outras linguagens de programação, "*free-source*", etc.

Os passos básicos para se usar o SWI-Prolog em seu ambiente interpretado são:

1. Editar o programa. Use o "*bloco de notas*" do windows, ou o "*edit*" do DOS. Preferencialmente editores ASCII;
2. Carregar o programa usuário para o ambiente interpretado SWI-Prolog;
3. Executar os predicados;
4. Validar os resultados;
5. Voltar ao passo inicial, se for o caso.

- Para carregar o programa: *pl* + <Enter>

```
Welcome to SWI-Prolog (Version 3.2.3)
Copyright (c) 1993-1998 University of Amsterdam.  All rights reserved.
For help, use ?- help(Topic). or ?- apropos(Word).
Yes
?-
```

- Para sair do ambiente SWI-Prolog:

```
?- halt.
```

- Para carregar um programa na Memória de Trabalho (MT) ("**Working Memory-WM**", ou "*base dinâmica de conhecimento*"):

```
?- consult('d:/curso/udesc/autind/iia/ex1').
d:/curso/udesc/autind/iia/ex1 compiled, 0.00 sec, 52 bytes.
Yes
?-
ou
?-  load_files('nomes','opcoes').
/* para conferir....se está na ``{\em Working  Memory}'' */
?- ensure_loaded(rosa).
Yes
?-
```

- Para eliminar um predicado da MT:

```
?- abolish(cor/1).
Yes
?- listing(cor/1).
[WARNING: No predicates for `cor/1']
No
?-
```

- Para executar uma regra:

```
?- eh_maior_que_10.
 DIGITE UM NUMERO:: 30
|    .  < ......... faltou o ponto . >
 numero maior que 10
Yes
?-
```

- Para "*bisbilhotar*" o "*help*", bem como inferir novas características do Prolog:

```
?- help(nome_do_predicado).
ou
?- apropos(padrão_desejado).
ou
?- explain(nome_do_predicado).
?- apropos(setenv).
setenv/2                        Set shell environment variable
unsetenv/1                      Delete shell environment variable
Yes
?-
```

Isto é, pretendendo fazer uma leitura do teclado, mas não conhece o predicado que faz tal função, procure os predicados relacionados com o "*apropos*", algo como: ?- apropos(read).

- As imperfeições do SWI-Prolog, ... *não pode existir espaço entre o nome do predicado e o início dos argumentos, no caso o "(...)".* Este erro é bastante comum entre os iniciantes do Prolog. Vejamos o exemplo abaixo no predicado ">":

```
?-    >     (9, 5).
[WARNING: Syntax error: Operator expected
> (9, 5
** here **
) . ]
?-   >(9, 5).
Yes
?-
```

- Para entrar no módulo de depuração ou debugger:

```
?- trace, predicado.
ou
?- debug, predicado.
ou
?- spy(predicado_em_particular).
para desativar:
?- nospy(predicado_em_particular).
e
?- notrace.
e
?- nodebug. /* ou Crtl-C para sair do modo debug */
```

1.11 Outros Detalhes

- Correções automáticas:

```
?- patos(X,Y).
Correct to: 'pratos(X, Y)'? yes
X = alho
Y = peixe ;        /*   ; ^ */
/*  ;  para forçar uma nova resposta manualmente */
X = cebola
Y = peixe ;
X = tomate
Y = peixe ;
No
?-
```

- Forçando todas as respostas, com o uso do predicado interno **fail**:

```
?- prato(X, Y),  write(X), nl, fail.
alho
cebola
tomate
No
?-
```

Logo o "*fail*" é um predicado que provoca uma falha na regra, ou seja, torna-a sempre falsa; esse "*fail*" força o mecanismo "*backtracking*".

- "*Apavoramentos*" com o modo "*trace*":

```
===> <abort.> sai do debug/trace ou da pendencia...
ou <Crtl C> +  <a> de abort
ou
?- abort.
Execution Aborted
Typing an 'h' gets us:
----
a:                abort      b:                break
c:                continue   e:                exit
g:                goals      t:                trace
h (?):            help
Action (h for help) ?
----
'What does this mean?', that's my first question.
'e'    exits the program immediately
'a'    aborts whatever you've type so far
'c'    continues where you left off
'h'    gets us the 'Action' help menu, again
So what does 'g', 't' and 'b' do?
I don't know yet, but I suspect:
'g'    lists possible goals ===> os pendentes
't'    traces the program, a debugging tool
'b'    breaks the current script (whatever your working on)
       and give a new shell.
```

- Não esqueça que:

```
?- trace, t.
e
?- spy(t), t.
```

são equivalentes !

Antes da próxima seção, sugiro que saibas a resposta do seguinte programa [1]:

[1] Este exemplo veio por sugestão de um aluno.

```
x(7).
x(5).
x(3).

par(Xpar) :- x(N1),  x(N2),  N1 =\= N2,   Xpar  is (N1+N2),
             write(N1) , write(' .... '),   write(N2),
               write(' .... '),  write(Xpar),
                nl,   fail.


?- par(N).
```

Cuja árvore de busca é dada pela figura 1.11:

Figura 3: Árvore de busca construída pelo Prolog para regra **par**

Aconselho estudar cada ponto deste exemplo, pois o mesmo sumariza todo conhecimento até o momento sobre Prolog. Avance à próxima seção, apenas se este exercício for dominado em detalhes. Use **guitracer** no XPCE, funciona muito bem e é gráfico.

2 Recursividade

A recursividade em Prolog é a sua própria definição em lógica. Apesar da ineficiência de soluções recursivas, esta é uma das elegâncias do Prolog. Os exemplos falarão por si.

A exemplo das demais linguagens de programação, uma função recursiva é aquela que busca em si próprio a solução para uma nova instância, até que esta encontre uma instância conhecida e retorne um valor desejado.

Exemplo: Calcular a soma dos primeiros n-valores inteiros:

$$S(n) = 1 + 2 + 3 + 4 + ... + (n-1) + n$$

Este problema pode ser reformulado sob uma visão matemática, mais especificamente, pela indução finita como:

$$S(n) = \begin{cases} 1 & \text{para } n = 1 \\ S(n-1) + n & \text{para } n \geqslant 2 \end{cases}$$

O que é um fato verdadeiro pois:

$$S(n) = \underbrace{1 + 2 + 3 + ... + (n-1)}_{S(n-1)} \quad +n$$

Como o procedimento é recursivo, temos encontrar a definição para "**parada**" da recursividade. Como n não tem limite superior, é para qualquer n, então vamos começar pelo que conhece:

#1. A soma de 1 é 1, logo: soma(1,1). **#2**. Para soma dos n-ésimos termos, é necessário a soma do $(n-1)$-ésimos termos, logo: soma(N,S) ... = ... Nant = (N-1), soma(Nant, S_Nant) e S = (N + S_Nant).

A regra #1, "*soma(1,1).*", é conhecida como condição ou *regra de parada*, ou ainda, *regra de aterramento*. Pelo fato de que o Prolog inicia seu procedimento sequencial "*de cima para baixo*", a condição #1 deve vir antes de #2. Em casos raríssimos, a regra recusirva, regra #2, vem antes da regra #1, contudo, há exemplos de tais casos. O exemplo acima reescrito em Prolog, é dado por:

```
s(1,1) :- true.                          /* regra #1 */
s(N, S) :- N > 1,                       /* regra #2 */
   Aux is (N-1),
   write('*'),
  s(Aux, Parcial),
  S is (N + Parcial).
```

O procedimento recursivo é típico da movimentação da pilha intrínseca do Prolog. O quadro abaixo ilustra o exemplo de um "*goal*" do tipo:

?-s(5,X).

Chamada Pendente	**Regra Casada**	**N**	**N1**	**Parcial**	**S**
s(5,X)	#2	5	4	?...<10>... →	...<15>...
s(4,X)	#2	4	3	?...<6>... →	↖...<10>...
s(3,X)	#2	3	2	?...<3>... →	↖...<6>...
s(2,X)	#2	2	1	?...<1>... →	↖...<3>...
s(1,X)	#1 (aterrada)	1	-	-	↖1

Duas observações:

1. A rigor, **internamente, a recursividade não procede exatamente como descrito no quadro acima** . Mas é algo muito "*próximo*" a isto;
2. Nem todo conjunto de regras recursivas são passíveis de admitirem tal quadro.

2.1 Exemplos sobre Recursividade

1. **Um simples sistema de menu**: além do uso da recursividade, este exemplo mostra o que fazer quando aparentemente não há predicados disponíveis ao que se deseja; ou seja: "$X <= 3$" não existe, mas como opção equivalente temos:

   - "$X =< 3$" ;
   - "$\+(X >= 3)$".

```
menu(0).
menu(_) :-
        repeat,
        write('......'), nl,
        ......................
        write('......'), nl,
        write(' DIGITE A OPCAO: '),
        read(X),
        X >= 0,
        \+(X >= 3),  /* isto é: X =< 3 */
        /*   X \== 0    é equivalente a:  \+(X == 3 ) */
        acao(X),
        menu(X).
%%%%%%%%%%%%%%%%%%%%%%%%%
        acao(0).     /* continua acoes etc */
        acao(1).
        acao(2).
        acao(3).
```

2. **Novamente o "Sócrates"**:

```
        human(socrates).          % facts about who is human
        human(aristotle).
        human(plato).
        god(zeus).                      % and who is a god
        god(apollo).
        mortal(X) :-human(X).   % a logical assertion that X is mortal if X is

Fazendo as perguntas:
        ?-mortal(plato).                % is Plato mortal?
        yes
        ?-mortal(apollo).               % is apollo mortal?
        no
        ?-mortal(X).                    % for which X is X mortal?
        X = socrates ->;
        X = aristotle ->;
        X = plato ->;
```

```
        no                                      %human

Evitando este ultimo "no", usando a recursividade:

        mortal_report :-
                write('Mostre todos mortais conhecidos:'), nl, nl,
                mortal(X),
                write(X), nl,
                fail.
       mortal_report.                 /* ou::    mortal_report :- true. */

Entao:
        ?- mortal_report.
        Mostre todos mortais conhecidos:'
        socrates
        aristotle
        plato
        yes
```

3. **Cálculo do fatorial**: Reformulando sob uma visão matemática, mais especificamente, pela indução finita temos:

$$Fat(n) = \begin{cases} 1 & \text{para } n = 0 \\ Fat(n-1) * n & \text{para } n \geqslant 1 \end{cases}$$

O que é um fato verdadeiro pois:

$$Fat(n) = \frac{\underbrace{1 * 2 * 3 * ..... * (n-1)}}{Fat(n-1)} * n$$

Como o procedimento é recursivo, temos de encontrar a definição para "**parada**" da recursividade. Como n não tem limite superior, é para qualquer n, então vamos começar pelo que conhece:

#**1**. O fatorial de 0 é 1, logo: fatorial(0,1). #**2**. O fatorial do n-ésimo termo, é necessário o fatorial do $(n-1)$-ésimo termo, logo: fatorial(N, Fat) :: Nant = (N-1), fatorial(Nant, Fat_Nant) e Fat = (N $*$ Fat_Nant).

em termos de Prolog temos:

```
fatorial( 0, 1 ).
fatorial( X, Fat ) :-
```

```
                        X  >  0,
                       Aux  is  (X - 1),
                       fatorial( Aux , Parcial ),
                       Fat  is  ( X   *  Parcial ).

Como eh apresentado por outro professor:
factorial(0, 1).
factorial(N, M):- N1 is N - 1,
                          factorial (N1, M1),
                          M is N*M1.

We can now read the Prolog code:
factorial(0, 1).   /* It is a fact that the factorial of 0 is 1*/
factorial(N, M) :- N1 is N - 1,   /* If temporary variable N1 is assigned to N-1*/
factorial(N1, M1),   /* and if the factorial of N1 is M1*/
M is N * M1.    /* and if M is assigned to N*M1*/
                        /* then the factorial of N is M    */
```

Complete a tabela abaixo, para descrição do fatorial:

Chamada Pendente	**Regra Casada**	**X**	**Aux**	**Parcial**	**Fat**
fatorial(5,X)	#	.....	.....	?..... →	.....
fatorial(4,X)	#	.....	.....	?..... →	↖.....
fatorial(3,X)	#	.....	.....	?..... →	↖.....
fatorial(2,X)	#	.....	.....	?..... →	↖.....
fatorial(1,X)	#	.....	.....	?..... →	↖.....
fatorial(0,X)	#	.....	.....	?..... →	↖.....

4. **Os ancestrais**: O ancestral de X, é no mínimo o pai de X. Logo, um avô X também é seu ancestral. O pai deste avô também é antepassado de X, e assim sucessivamente. Generalizando este conceito de ancestralidade, veja figura 4, em termos da relação *pai*'.

Figura 4: Uma árvore que representa os descendentes de alguém

Escrevendo este conceito em Prolog, temos:

```
ancestral(X,Y) :- pai(X,Y).
ancestral(X,Y) :- pai(X,Z), ancestral(Z,Y).
```

5. "**O caminho do português**": A proposta do problema é encontrar o custo entre dois vértices quaisquer em um grafo orientado dado pela figura 5. O nome do problema define a proposta do exercício. Este grafo não apresenta ciclos, e nem é bidirecional entre os vértices.

Figura 5: Um grafo que define custos entres as cidades

```
ligado(a,b,5).      ligado(a,c,10).     ligado(a,g,75).
ligado(c,d,10).    ligado(d,g,15).     ligado(d,e,5).
ligado(g,f,20).    ligado(e,f,5).       ligado(b,f,25).
ligado(b,e,5).    ligado(b,f,25).

rota(X,Y,C) :- ligado(X,Y,C).
rota(X,Y,C) :- ligado(X,Z,C1),
                      rota(Z,Y, C2),
                      C is (C1 + C2).

?- rota(a,g,X).
X = 75 ;
X = 35 ;
No
?-
```

3 Functores

3.1 Definições

Functores: são funções lógicas, logo a equivalência entre um domínio e imagem é identificada. Os functores permitem contornar o mapeamento típico sobre $\{V, F\}$ ("*yes*" ou "*no*", conforme o aluno já tenha percebido); extrapolando com isso a um domínio de objetos mapeados à objetos.

Exemplo: seja a descrição de uma classe de objetos do tipo carro (ver figura 3.1):

```
carro(Nome, estilo(
        esporte(Portas, opcionais(L1,L2)),
        passeio(Portas),
        off_road(L3, motor(Comb, outros(Turbo, L4)))
                              )    ).
```

Figura 6: Representando o conceito de hierarquia com um functor

3.2 Motivação

A principal observação do exemplo acima é a hierarquica "*natural*" e estabelecida, para representar um conhecimento. Tal representação é conhecida por "*frames*", que incorpora a maioria dos conceitos da programação orientada-à-objetos. Em resumo temos:

1. Construção de fatos genéricos (próximo ao conceito de classes, instâncias, etc.);
2. Quantidade variável de argumentos, leva há novos predicados. Há com isso *um certo ganho* de generalidade, pois um mesmo predicado pode ser utilizado para descrever vários tipos de objetos, buscas sobre padrões são facilitadas;
3. Legibilidade dos dados, e consequentemente o entedimento do problema;
4. Em resumo, os functores podem serem vistos como uma estruturação os dados sob formas dos "*REGISTROS*" convencionais.

3.3 Exemplo

Seja o programa abaixo

```
artista("Salvador Dali", dados(pintor(surrealista), regiao( "Catalunha" ), 86, espanhol)).
artista("Pablo Picasso", dados(pintor(cubista), 89, espanhol)).
artista("Jorge Amado", dados(escritor(contemporaneo), 86, brasileiro)).
```

Para verificar seu aprendizado, verifique e entenda o porquê das perguntas e repostas abaixo:

1. Dentro do predicado "*artista*" há outros predicados?
   Resp: Errado... não são outros predicados ... e sim **funções lógicas**!
2. Quantos argumentos tem o predicado artista?
   Resp: 02 argumentos, um campo é o "'*nome*"... e o outro é a função "*dados*" !
3. Quantos argumentos tem a função "*dados*"?
   Resp: 04 e 03 argumentos para o primeiro e segundo fato respectivamente. Logo, apesar de terem o mesmo nome, elas são diferentes!
4. Quantos argumentos tem a função "pintor"?
   Resp: 01 argumento!
5. A função "*pintor*" é equivalente a "*escritor*"?
   Resp: Não, um objeto tem características da função "*pintor*" e outro da "*escritor*";
6. Quanto ao valor de retorno dessas funções lógicas?
   Resp: Não existe... **ela por si só já é mapeada num domínio qualquer (números, letras, objetos, etc.....)**.

3.4 Discussão sobre Problemas de IA

Reescrevendo sob a forma regras o problema dos vasos (falta a listagem completa, que é apresentada em sala de aula):

```
    ................................................
    vaso(4,Y) :- vaso(X,Y) , X < 4.
```

```
.....................................................
conjunto de regras de movimentacao de enchimento
e esvaziato. dos vasos...
.....................................................
vaso(W,Z) :- vaso(X,Y), X+Y >= 4, Y > 0, Z is (Y-(4-X)),
         W is 4. /* enche o vaso A com sobra em B */
.....................................................
vaso(2,_) :- vaso(_,2).
.....................................................
```

Perguntas e Problemas:

1. Todas as regras tem que serem explicitadas?
   Resp: SIM.... são possibilidades de *movimentos* e/ou *ações* plausíveis de serem acionadas! Ex: despejar, encher, esvaziar... etc.
2. Se tudo for explícito ... logo tenho a solução do do problema?
   R: Não ... pois se tivéssemos os caminhos, ou pelo menos um, o problema seria "*algoritimizável*", portanto, livre da IA. Mas mesmo encontrando uma solução, isto ainda não garante que esta seja uma solução robusta. Pois, com uma modificação da condição inicial, ou de alguma alteração do problema, todo algorítmo "*cairia por água abaixo*". Um algorítmo funciona para uma condição estática e esperada, algo que difere de programas de IA, que são genéricos para uma classe de situações/instâncias de um problema.
3. Afinal, qual a proposta da IA em termos de ferramentas para resolver problemas?
   Resp: A idéia é que estas ferramentas (ver livro texto), no caso uma linguagem de programação do tipo Prolog, use um mescanismo "*inteligente*" de tentar soluções. Ou seja, diferentemente as Linguagens tradicionais, onde problema tinha que ser descrito em termos de sua solução, no Prolog a proposta é apenas "*declarar o problema*"! Logo, o Prolog é uma linguagem *declarativa*, e não *procedural* como as demais. Sendo assim, através de fatos e regras descreve-se o problema, e é deixado ao seu sistema de dedução interno buscar às soluções possíveis!
4. A representação acima é apropriada?
   Resp: Em termos de RC (Representação do Conhecimento) a resposta é "**sim**"! Em termos de Prolog, não, pois de imediato temos os seguintes problemas:
   (a) As regras são tipicamente recursivas. Nesse momento, tal aspecto é INDESEJÁVEL! A recursividade é elegante (e muito na maioria dos casos), contudo, computacionalmente, a nível de uso/gasto de memória, é ineficiente.
   (b) Decorrente do item anterior, podem surgir laços inifinitos ("*loops*");
   (c) Não há controlabilidade de quais regras foram usadas ou não ! Pois, para o Prolog todas as regras são factíveis de serem disparadas. Óbvio que é um resultado esperado e desejável.

(d) Então precisamos de estruturas mais poderosas que functores?
Sim, vamos usar um tipo especial de functor chamado de "*lista*". As listas em Prolog servem como um poderoso mecanismo de armazenamento temporário. No caso, o interesse é sobre o armazenamento de regras já disparadas e estados já visitados !

3.5 Usando Functores em Problemas de IA

Retomando o ponto acima, vamos contornar em parte o problema. Seja:

```
    vaso(W,Z)  :- vaso(X,Y), X+Y >= 4, Y > 0, Z is (Y-(4-X)),
                        W is 4. /* enche o vaso A com sobra em B */

    como reescreve-la usando o conceito de functores:

 movimento(X,Y  --->  W,Z) :- X+Y >= 4, Y > 0, Z is (Y-(4-X)), W is 4.
 movimento(vaso(X,Y), vaso(W,Z)) :- X+Y >= 4, Y > 0, Z is (Y-(4-X)), W is 4.

        Estado anterior

                              Novo estado...ou corrente !
```

3.6 Concluindo Functores

Alguns de regras e questões que podem ser encontrados a partir do exemplo acima:

```
 /*  qual o nome e os dados de cada artista ? */
?- artista(X,Y), write(X), nl, write(Y),nl, fail.

 Do exemplo, deduzimos que functor ``{\em casa}'' (``{\em
matching}''  com variáveis (letras maiusculas)....

/* quem são os cubistas ?*/
?- artista(X, dados(pintor(Y) ,_ ,_ )), Y == cubista, nl,
     write(X), write('=====>'), write(Y), nl, fail.

/* quem tem mais de 80 anos ? */
?- artista(X,dados(_ , Y _)), Y > 80, nl, write(X),  write(Y),nl, fail.

/* no caso acima, nem todos resultados foram encontrados, porquê ? */
```

Resumindo: os functores organizam dados e visualizam regras recursivas de uma maneira mais simples e controlável. Outro detalhe é que como objetos, functores respeitam todas as regras de "*casamento*" vistas até o momento. Sem exceção !

4 Listas

4.1 Definições

Definições iniciais:
- Uma lista é uma estrutura de dados que representa uma coleção de objetos homogêneos;
- Uma lista é uma seqüência de objetos;
- Uma lista é um tipo particular de functor[2] (veja esta nota de roda-pé), pois apresenta uma hierarquia interna.

Notação: O símbolo "[" é usado para descrever o início de uma lista, e "]" para o final da mesma;

Exemplos: [a, b, c, d], logo um predicado cujo argumento seja algumas letras, temos uma lista do tipo:

```
letras( [ a, b, c, d ] ).
          ^
          |
          |
  cabeça da lista
```

Os elementos de uma lista são lidos da esquerda para direita, logo a letra "a" é o primeiro elemento ou "*cabeça*" da lista. Quanto ao resto ou "*cauda*" da lista, é uma "*sub-lista*" dada por: [b, c, d]. Esta sub-lista, de acordo com uma outra definição complementar apresentada na seção 4.2, também é uma lista.

Operador |: "*Como vamos distinguir de onde se encontra a cabeça da cauda da lista?*" Como o conceito de listas introduziram novos símbolos, isto é, os seus delimitadores [...], há um novo operador que separa ou define quem é a cabeça da cauda da lista. Este operador é o "*pipe*", simbolizado por:|, que distingue a parte da esquerda da direita da lista. Isto é necessário para se realizar os casamentos de padrões com as variáveis.

Exemplos de "*casamentos*": os exemplos abaixo definem como ocorrem os casamentos entre variáveis e listas. Portanto, preste atenção em cada exemplo, bem como teste e faça suas próprias conclusões de como as listas operam.

[2] **Logo, a lista vai apresentar uma hierarquia natural, internamente.**

```
[ a, b, c, d ] == X
[ X | b, c, d ]  ==  [ a, b, c, d ]
[ a | b, c, d ]  ==  [ a, b, c, d ]
[ a , b | c, d ]  ==  [ a, b, c, d ]
[ a , b , c | d ]  ==  [ a, b, c, d ]
[ a , b , c , d | [] ]  ==  [ a, b, c, d ]
[] == X
[ [ a | b , c , d] ] == [ [ a , b , c , d ] ]
[  a | b , c , [ d ] ] == [  a , b , c , [ d ] ]
[  _ | b , c , [d ] ] == [  a , b , c , [ d ] ]
[  a | Y ] == [  a , b , c ,  d ]
[  a | _ ] == [  a , b , c ,  d ]
[  a , b | c , d ] == [  X , Y | Z ]
```

Contra-exemplos de "*casamentos*": Explique porque nos exemplos abaixo, **não** ocorre o casamento de padrões:

```
[ a , b | [c, d] ]  \==  [ a, b, c, d ]
[ [ a , b , c , d] ]  \==  [ a, b, c, d ]
[  a , b , [ c ] , d, e ]  \==  [ a, b, c, d, e ]
[ [ [ a ] | b , c , d] ] \== [ [ a , b , c , d] ]
```

Enfim, os tipos de casamentos de objetos de uma lista, segue o mesmo padrão dos "*matching*" considerados até o momento em Prolog.

Aplicação: Devido ao fato de listas modelarem qualquer estrutura de dados, invariavelmente, seu uso é extensivo em problemas de IA, pois envolvem buscas sobre estas estruturas.

4.2 Exemplos

Retomando ao conceito de listas, se "*auto-define*" o conceito de lista com os seguintes axiomas:

1. "*Uma lista vazia é uma lista*";
2. "*Uma sub-lista é uma lista*".

As definições acima são recorrentes, isto é, uma depende da outra. Em outras palavras, temos uma definição recursiva uma vez mais. Sendo assim, reescrenvendo em Prolog tal definição é dada por:

```
#1          é_uma_lista( [ ] ).                        /* 1a. premissa */
#2          é_uma_lista( [X | T ] ) :- é_uma_lista( T ).    /* 2a. premissa*/

?- é_lista( [a,b,c] ).
yes
```

Um "*mapa de memória aproximado*" é dado por:

	Regra	X	T
é_uma_lista([a,b,c])	#2	a	[b,c]
é_uma_lista([b,c])	#2	b	[c]
é_uma_lista([c])	#2	c	[]
é_uma_lista([])	#1	–	–

Basicamente, quase todas operações com listas possuem regras análogas a definição acima. O exemplo anterior serve apenas para identificar que o objeto: [a,b,c,d], é uma lista. Contudo, regras sobre listas são inúmeras e bastante elegantes, algumas são mostradas nos exemplos que se seguem:

1. **Comprimento de uma lista**:

```
#1        compto([ ], 0).
#2        compto([X | T], N):- compto(T, N1), N is N1+1.

? - compto([a, b, c, d], X).
X = 4
```

Um "*mapa de memória aproximado*" é dado por:

	Regra	X	T	N1	N is N+1
compto([a,b,c,d],N)	#2	a	[b,c,d]	3 →	3+1=4
compto([b,c,d],N)	#2	b	[c,d]	2 →	↖ 2+1
compto([c,d],N)	#2	c	[d]	1 →	↖ 1+1
compto([d],N)	#2	d	[]	0 →	↖ 0+1
compto([],N)	#1	–	–	–	↖ 0

2. **Concatenar ou união de duas listas**: Em inglês este predicado[3] é conhecido como "*append*", e em alguns Prolog's pode estar embutido como predicado nativo:

```
#1  uniao([],X,X).
#2  uniao([X|L1],L2,[X|L3]) :- uniao( L1, L2, L3).

O ‘‘goal’’:
?- uniao([a,c,e],[b,d], W).
W=[a,c,e,b,d]
yes
```

[3]A palavra predicado, neste contexto, reflete o conjunto de regras que definem as operações dos mesmos sobre listas

Um "*mapa de memória aproximado*" é dado por:

	Regra	X	L1	L2	L3	L≡[X | L3]
uniao([a,c,e],[b,d],L)	#2	a	[c,e]	[b,d]	*[c,e,b,d]*	*[a,c,e,b,d]*
uniao([c,e],[b,d],L)	#2	c	[e]	[b,d]	*[e,b,d]*	↖ *[c,e,b,d]*
uniao([e],[b,d],L)	#2	e	[]	[b,d]	*[b,d]*	↖ *[e,b,d]*
uniao([],[b,d],L)	#1	–	–	[b,d]	–	↖ *[b,d]*

3. **Dividir uma lista em duas outras listas**: Lista inicial é "em [X,Y | L]", em uma lista

```
#1  divide([], [], []).          % num par de objetos na lista
#2  divide([X], [], [X]).          % num impar de objetos na lista => lista L2
#3  divide([X,Y | L3] , [X | L1], [Y | L2] ):-
                                                divide( L3, L1, L2).
```

Obs: Estes dois últimos predicados apresentam uma particularidade interessante. Permitem que os predicados encontrem a lista original. Exemplo:

```
?- divide([a,b,c,d,e],L1,L2).
L1=[a,c]      L2=[b,d,e]
?- divide(L , [ a , b ], [ c , d ]).
L=[a, c, b, d]
```

Um "*mapa de memória aproximado*" é dado por:

	Regra	X	Y	[X | L1]	[Y | L2]	L3
divide([a,b,c,d,e],L1,L2)	#3	a	b	*[a,c]*	*[b,d,e]*	[c,d,e]
divide([c,d,e],L1,L2)	#3	c	d	*[c]*	*[d,e]*	[e]
divide([e],L1,L2)	#2	e	–	*[]*	*[e]*	–

4. **Imprimir uma lista**: observe o uso do predicado "*put*" ao invés do "*write*". Esta troca se deve a razão que o Prolog trata as listas no código original ASCII, ou seja "fred" = [102,101, 114, 100].

```
escreve_lista( [ ] ).
escreve_lista( [ Head | Tail ] ) :-
                                        write( ' : ' ),
                                        put( Head ),
                                        escreve_lista( Tail ).
```

5. **Verifica se um dado objeto pertence há uma lista**: novamente, em alguns Prolog's, este predicado pode estar embutido, confira.

```
member( H, [ H |  _ ] ).
member( H, [ _  | T ] ) :- member(H, T).
```

6. **Adiciona um objeto em uma lista, caso esse não esteja contido**:

```
add_to_set(X, [ ], [X]).
add_to_set(X, Y, Y) :- member(X, Y).
add_to_set(X, Y, [X | Y]).
```

7. **O maior valor de uma lista**: retornar o maior valor de uma lista.

```
1.    max( [] ,0) :- !.
2.    max( [M] , M ) :- !.
3.    max( [M , K], M ) :- M >= K , !.
4.    max( [M|R] ,N ) :- max( R , K ) , max( [K , M] , N).
```

   Neste momento, o aluno deve ter percebido que a ordem com que as regras se encontram dispostas na Memória de Trabalho (MT) deve ser considerada. Pois o mecanismo de "*backtracking*" do Prolog, força uma ordem única de como estas regras serão avaliadas.

   Sendo assim, uma disciplina de programação em Prolog se faz necessária algumas vezes! **No caso das listas, as condições de paradas devem ocorrer antes da regra geral recursiva**. A exceção é a leitura de uma lista, este predicado é descrito mais adiante. A regra geral é aquela que contém uma recursividade, no exemplo anterior, é a regra número 4. As demais regras, 1 a 3, são regras destinadas à parada da recursão. Logo, obrigatórias.

   Estas regras de parada, também são chamadas de regras ou cláusulas "*aterradas*" ("*grounding*", pois delimitam o final da recursão. Como exercício, determine nos exemplos anteriores quem são as regras gerais e as aterradas.

8. **Inverter uma lista**: este método é ingênuo (primário) na inversão de uma lista, no sentido que faz $n(n+1)/2$ chamadas recursivas a uma lista de comprimento n.

```
naive_reverse( [ ] , [ ] ).
naive_reverse( [ H  | T ], Reversed ) :-
        naive_reverse( T , R ),
        append( R , [ H ], Reversed ).
```

9. **Inversão sofisticada de uma lista**: usa como "*truque*" um acumulador, compare com o anterior.

```
xinvertex(A, Z) :- reverse(A, [ ], Z).
reverse( [ ] , Z, Z).
reverse( [A | X ], Acumulador, Z ) :-
                            reverse( X, [ A | Acumulador ],  Z ).
```

10. **Leitura de uma lista via teclado**: observe que a cláusula aterrada "**quase sempre**" se encontra antes da cláusula geral. Contudo, a leitura de uma lista é uma das raras exceções em que o aterramento vem depois da regra geral recursiva.

```
le_lista( Lista ) :-
                    write('Digite <Enter> ou <Escape> para terminar: '),
                    write('  ===> '),
                    le_aux( Lista ).
le_aux( [Char | Resto ] ) :-
                                        write('  '),
                                        get0(Char),
                                        testa(Char),
                                        put(Char),
                                        put(7),  /* beep */
                                        le_aux( Resto ).
/* Condição da Parada */
      le_aux( [ ] ) :- !.
testa(13) :- !, fail.        /* Return */
testa(10) :- !, fail.        /* New line ===> UNIX */
testa(27) :- !, fail.        /* Escape */
testa( _ )  :- true.
```

4.3 Combinando os Functores as Listas

Os problemas de IA, basicamente se utilizam do núcleo descrito abaixo, ou uma variação sobre o mesmo. Acompanhe a discussão com o professor, bem como os exemplos resolvidos no capítulo de Resolução de Problemas.

```
/* Exemplo do uso de Functores na Resoluçao de Problemas
   com Prolog*/


estado( ((X,Y,Z) ,W) ) :- W is (X mod 4) , regras_de_estado( W, (X,Y,Z) ).

/* regras ... condicionais do problema.... */
regras_de_estado( 0, (X,Y,Z) ):-  Y =\= 0, Z is (X / Y). /* a divisao */
regras_de_estado( 0, (_,0,0) ).   /* se denominador Y for 0... retorne 0) */
regras_de_estado( 1, (X,Y,Z) ) :- Z is (X * Y).
regras_de_estado( 2, (X,Y,Z) ) :- Z is (X + Y).
regras_de_estado( 3, (X,Y,Z) ) :- Z is (X - Y).
/* ... essas representam as condicoes de contorno do problema,
  restricoes... e permissoes... declaradas */
```

```
cod(0, divide).
cod(1, mult).
cod(2, soma).
cod(3, subtrai).

inicio :-
     busca([((0,0, _), _ )], Solution),
     reverse(Solution, ReversedSolution), /* inverte a lista */
     imprime(ReversedSolution).

imprime(X) :- w_l(X).
/* uma saida formatada */
w_l([]).
w_l([((X,Y,Z),W)|L]) :-
nl,
cod(W, Op),
write(X), write(' .. '), write(Op), write(' .. '), write(Y), write('....  é: '),
write(Z),
w_l(L).

/* Núcleo do processo de busca...*/
busca( [ ((X,Y,Z), W1_op) | Resto ], Solution) :-

     estado( ((X,Y,Z),W1_op) ),
     Xnovo is  (X + 1),
     Ynovo is  (Y + 1),
     Xnovo =< 17,
     Ynovo =< 17,
     estado( ((Xnovo,Ynovo,Znovo),W2_op) ),

     \+member( ((Xnovo,Ynovo,Znovo),W2_op) , Resto), /* não é menbro */
     busca([((Xnovo,Ynovo,Znovo),W2_op),((X,Y,Z),W1_op)|Resto], Solution).

/*
 Condição de Parada, um estado final... 17
*/

busca([((17,17, X),W) | Resto],[((17,17, X ),W) | Resto]).

/*  vem o primeiro argumento de lista, e volta a mesma como parada */
```

5 Gerando Programas Executáveis (ou quase com o SWI-Prolog)

O Prolog gera executáveis com velocidades compatíveis a linguagem C++, Delphi, e outras. Contudo, o SWI-Prolog pelo fato de ser um "*freeware*" (talvez), gera executáveis que traz consigo o seu interpretador. Ou seja, um executável no SWI-Prolog funciona como um Java, na base de um "*bytecode*". Caso queiras desenvolver aplicações comerciais ou de alto-desempenho, sugiro comprar um Prolog de algum dos muitos fabricantes, como www.prologcenter.com.

A seguir é mostrado como gerar um "*executável*" com o SWI-Prolog. Considere um programa exemplo, como este:

```
x(1).
x(5).
x(3).
par(PAR) :- x(N1),
            x(N2),
          N1 =\= N2,
          is(PAR , (N1+N2)),
        write(PAR), write(' .... '), write(N1), write(' .... '),
        write(N2), nl, fail.
/*, fail. */
par( 0 ) :- true.

inicio :- nl, par(_),  halt.
```

Construa um dos dois scripts (isto é Linux) que se seguem:

```
#!/bin/sh
base=~/pesquisa/livro/prolog/pgms
PL=pl
exec $PL -f none -g "load_files(['$base/impares'],[silent(true)])" \
        -t inicio -- $*
```

ou

```
#!/bin/sh
pl --goal=inicio --stand_alone=true -o saida.exe -c impares.pl
```

Execute um destes scripts na linha de comando do Linux, conforme a ilustração abaixo:

```
[claudio@goedel pgms]$ sh comp1.script

6 .... 1 .... 5
4 .... 1 .... 3
6 .... 5 .... 1
8 .... 5 .... 3
4 .... 3 .... 1
8 .... 3 .... 5

ou

[claudio@goedel pgms]$ sh comp2.script
% impares.pl compiled 0.00 sec, 1,524 bytes
[claudio@goedel pgms]$ ./saida.exe

6 .... 1 .... 5
4 .... 1 .... 3
6 .... 5 .... 1
8 .... 5 .... 3
4 .... 3 .... 1
8 .... 3 .... 5
[claudio@goedel pgms]$
```

Neste último caso um executável foi gerado chamado "*saida.exe*".

Estes dois scripts são é equivalentes ao seguinte procedimento dentro do ambiente interpretado:

```
?- consult('impares.pl').
% impares.pl compiled 0.00 sec, 1,556 bytes

Yes
?- qsave_program('nova_saida.exe', [goal=inicio, stand_alone=true, toplevel=halt]).
Yes
```

Aqui, o predicado "*qsave_program*" gerou um executável chamado de "*nova_saida.exe*". Leia com atenção o *help* deste predicado.

A interface com linguagens como C, Java e outros é relativamente fácil. Contudo, neste momento será omitido.

6 Operações especiais

Observe o exemplo que se segue:

```
?- dynamic(algo_novo/1).
Yes
?-
?- assert(algo_novo(  alo___mundo  )).
Yes
?-
?- algo_novo(X).
X = alo___mundo ;
Yes
?-
```

Basicamente, o que se precisa fazer para remover ou adicionar um fato ou regra à MT, bastam dois passos:

1. Definir que este predicado é dinâmico na MT. Use o predicado: **dynamic(nome_do_novo/aridade)**.
2. Pronto para usar, com o **assert** ou **retract**.

Veja outros exemplos:

1. Removendo uma regra ou fato da Memória de Trabalho (MT):

   ```
   ?- retract(uniao([A|B], C, [A|D]) :- uniao(B, C, D)).
   ```

2. Adicionando **uma regra** ou **fato** da MT:

   ```
   ?- asertz(uniao(B, [A|C], [A|D]) :- uniao(B, C, D)).
   Correct to: ‘assertz( (uniao(B, [A|C], [A|D]):-uniao(B, C, D)))’?
   yes
   B = _G519
   A = _G513
   C = _G514
   D = _G517
   Yes
   ```

3. Finalmente, reusando um fato já adicionado:

   ```
   ?- assert('uma_regra(X) :- algo_novo(X).').
   Yes
   /* usando a regra recém incluída */
   ?- uma_regra(Y).
   Y = alo___mundo ;
   Yes
   ?-
   ```

Enfim, avalie o impacto do que representa **incluir** ou **excluir** uma regra durante a execução de um programa. Ou seja, um programa que se "*auto-modifica*" durante a sua execução!

7 Programando com "*Elegância*"

Há um estilo de como programar **bem** em Prolog? Claro, a resposta é um **sim**. Contudo, esta elegância ou estilo de programação não é trivialmente expresso sob a forma de regras. Algumas dicas (experiências diversas) são:

- Entenda **profundamente** o problema que queres escrever em Prolog. Um problema mal entendido, dificilmente será bem implementado (se é que for);
- Escreva da mesma maneira que o problema é montado mentalmente. Assim como se fala, se escreve em Prolog. "**Declare o problema, e não o resolva**", esta é uma das máximas do Prolog;
- Evite o uso de operadores tradicionais como: $>$, $<=$, *is* ... etc, isto normalmente revela uma proximidade com a programação procedural. O foco do Prolog é "**Casamento de Padrões**". Pense que dois objetos podem ser equivalentes ou diferentes, apenas isto. Logo, eles casam ou não;
- Quando uma implementação começa ficar complicada, é porque alguma assumida previamente, está errada. Volte atrás, e refaça tudo sob um outro enfoque. Refazer um novo programa por inteiro, normalmente é mais simples do que "*forçar*" um problema mal estruturado, convergir em uma solução aceitável;
- Consulte os "*grandes mestres*" em Prolog. Aprender Prolog com programadores experientes é uma boa dica. Alguns conhecidos no Brasil: Pedro Porfírio (porfirio@unifor.br), Edilson Ferneda (ferneda@pos.ucb.br). Finalmente, a lista de discussão do SWI-Prolog também é fácil encontrar muitos peritos em Prolog.

8 Gerando Sequências de Dominó

Este exemplo é ilustrativo, mas que pode ser aplicado há problemas de vários domínios. Principalmente, os problemas clássicos de IA. Adicionalmente, este exemplo aplica todos os conceitos vistos nesta apostila, inclusive functores. Observe atentamente o código, e acompanhe a discussão em sala de aula.

```
/* Exemplo do uso de Functores na Resoluçao de Problemas
   com Prolog ==> Dominó de um lado só....*/

/* por uma peça... pedra... no domino....*/

muda_estado( (X,Y) , (Z,W) ) :-  pc( (Z,W ) ),  Y == Z, X =\= W.
muda_estado( (X,Y) , (W,Z) ) :-  pc( (Z,W ) ),  Y == W, X =\= Z.

/* esta ultima condicao.... evita que a mesma peca seja usada....duas veze */
```

```
/* a ponta livre Y, encaixa com um dado Z ... tem que ser uma peça livre...*/

/* regras ... condicionais do problema.... */

/* ... essas representam as condicoes de contorno do problema,
  restricoes... e permissoes... declaradas
*/

pc( (1,1) ). pc( (1,2) ).  pc( (1,3) ). pc( (1,4) ).
pc( (2,2) ). pc( (2,3) ).  pc( (2,4) ).
pc( (3,3) ). pc( (3,4) ).
pc( (4,4) ).


inicio :-
     busca([ (1,1) ], Solution),
     reverse(Solution, ReversedSolution), /* inverte a lista */
     imprime(ReversedSolution),
     write('\n*************************').


tudo  :-
findall( Solucao, busca([ (1,1) ], Solucao), Todas_Sol),
elimina_duplicados( Todas_Sol,Lista_solucoes),
/* reverse(Lista_solucoes, Lista_solucoes_r), */
imp_l(Lista_solucoes),
nl,
write(' Número de solucoes é :'),
comp(Lista_solucoes, T),
write(T).


elimina_duplicados( [],[] ).
elimina_duplicados( [X|L1] , [X|L2]) :-
       \+pertence( X , L1 ),
elimina_duplicados( L1 , L2).
elimina_duplicados( [ _ |L1] , L2 ) :-
elimina_duplicados( L1 , L2).

pertence( X, [X|_]).
pertence( X, [_|L]) :-  pertence( X, L ).


imp_l([]).
imp_l([X|L]) :- nl,
/* write(X), */
reverse(X,Y),
```

```
        w_l(Y),
imp_l(L).


imprime(X) :- w_l(X).
/* uma saida formatada */
w_l([]).
w_l([(X,Y)|L]) :-
nl,
write('----------------'), nl,
write('  |  '),write(X),write('  |  '), write(Y),write('  |'),
w_l(L).

/*
 Condição de Parada, um estado final... sem peças... ou sem casamentos...
*/

busca(L,L) :-  comp(L,X) , X == 5.
/* >=( X , 4 ).   */
/* =<( X , 4 ). */


/* observe que Solution da chamada anterior ... é instanciada por casamento...
   nada de atribuições...
*/


/* Núcleo do processo de busca...*/
busca( [ (X,Y) | Resto ], Solution) :-

     muda_estado(    (X,Y),  (Xnovo,Ynovo)   ),
     /* estados/peças novos   */

     \+pertence_jogo( (Xnovo,Ynovo) , Resto),
     /* se esta peça não é menbro do jogo ainda... */
     busca([(Xnovo,Ynovo) , (X,Y) | Resto], Solution).



comp([],0).
comp([_|Resto],Total):- comp(Resto,P_resto), Total  is (P_resto +1).

/*  ve se o primeiro argumento de lista, pertence a ela...e volta como parada: yes */
pertence_jogo( (X,Y), [(X,Y) | _]).
pertence_jogo( (X,Y), [(Y,X) | _]).
pertence_jogo( (X,Y), [ _ | Resto]) :-  pertence_jogo( (X,Y), Resto ).
```

9 Sites interessantes

- http://www.cbl.leeds.ac.uk/ tamsin/Prologtutorial/
- http://www.sju.edu/ jhodgson/ugai/
- http://www.cee.hw.ac.uk/ alison/ai3notes/
- http://www.sics.se/ps/sicstus.html
- http://dobrev.com/download.html
- http://www.swi-prolog.org/
- http://swi.psy.uva.nl/projects/xpce/download.html
- http://www.amzi.com (tem vários artigos e tutoriais que ilustram o uso e aprendizado do Prolog, respectivamente. Um site fortemente recomendado, pois há um farto material gratuito, incluindo um ambiente de programação.)
- http://www.arity.com (http://www.arity.com/www.pl/products/ap.html tem um outro Prolog free)
- http://www.ida.liu.se/ ulfni/lpp/
- Strawberry Prolog http://www.dobrev.com/

Finalizando, alguns bons livros sobre Prolog:

- Michael A. Convigton, Donald Nute, André Vellino; *Prolog - Programming in Depth*, Prentice-Hall, 1997;
- Ivan Bratko; *Prolog, Programming for Artificial Intelligence*, 2nd edition (or later if there is one), Addison-Wesley;
- W.F. Clocksin and C.S. Mellish; *Programming in Prolog*, 3rd edition, Springer-Verlag;
- Leon Sterling and Ehud Shapiro; *The Art of Prolog*, MIT Press;
- Richard A. O'Keefe; *The Craft of Prolog*, MIT Press 1990;
- Há um título de um livro de IA, que é "*aproximadamente*" é: "*Solving Complexs Problems with Artificial Intelligence*", cuja linguagem utilizada nas soluções, só poderia ser em Prolog!